

Simulation von Wärmeleitungsvorgängen

von Reinhard Oldenburg, Frankfurt

Durch die Zerlegung in diskrete Zeitschritte werden mit Hilfe von Tabellenkalkulation Wärmeleitungsprozesse modellierbar, die traditionell partielle Differentialgleichungen erfordern.

1 Einleitung

Simulation ist selbst gemachte Realität. Um die Sicherheit von Autos zu verbessern, das Wetter vorherzusagen oder nur um die Fallzeit eines Gegenstandes zu berechnen, benutzt man Simulationsprogramme. Man kann sie immer dann einsetzen, wenn man die grundlegenden (physikalischen, chemischen,...) Gesetzmäßigkeiten kennt. Im Wesentlichen besteht die Aufgabe der Simulationsprogramme dann darin, die Gleichungen eines mathematischen Modells zu lösen und ggf. grafisch darzustellen.

Die Leistungen moderner Simulationsprogramme sind beeindruckend. Es ist eine Aufgabe des Mathematikunterrichts den Schülern die Möglichkeit zu bieten, zu erleben, wie gut sich auch komplexe Prozesse der Realität simulieren lassen. Dies trägt in doppelter Weise zur Entwicklung des Weltbildes der Schüler bei, einerseits mit der Erkenntnis, dass die Realität weitgehend (wenn natürlich auch nicht umfassend) simulierbar ist (eine Erkenntnis über die Gesetzmäßigkeit der Welt), und andererseits mit dem Wissen um die kulturelle Leistung, die sich in dieser Fähigkeit ausdrückt.

In diesem Beitrag wird unter den vielen verschiedenen Realsituationen, die sich in der Schule simulieren lassen, die Wärmeleitung ausgewählt. Zwar mag die Simulation von Bewegungsvorgängen von Körpern nahe liegender erscheinen, aber die Wärmeleitung ist intuitiv noch zugänglicher und es besteht weniger Kollisionsgefahr mit den Inhalten des Physikunterrichts.

2 Ein heißer Klotz

Ein Stück Eisen wird in kochendem Wasser auf 100°C erhitzt oder ein Gefäß wird

mit heißem Wasser gefüllt und dann auf den Tisch gestellt. Klotz oder Flüssigkeit kühlen sich ab. Die Erfahrung (entweder aus der Erinnerung oder einem einfachen Experiment) zeigt: Die Temperatur gleicht sich der Umgebungstemperatur an und die Geschwindigkeit der Anpassung ist anfangs – also bei großer Temperaturdifferenz – hoch, später dann kleiner. Eine Messung – hier eine Kaffeetasse statt einem Klotz – bestätigt dies.

| Zeit/Min | Temp/ |
|----------|-------|
| 0 | 76 |
| 1 | 73 |
| 2 | 70 |
| 3 | 68 |
| 4 | 66 |
| 5 | 64 |
| 6 | 62 |
| 7 | 60 |
| 8 | 59 |
| 9 | 57 |
| 10 | 56 |
| 11 | 55 |
| 12 | 53 |
| 13 | 52 |
| 14 | 51 |
| 15 | 50 |
| 16 | 50 |
| 17 | 49 |
| 18 | 48 |
| 19 | 47 |
| ... | ... |
| 100 | 22 |

Die Temperatur T ist eine Funktion $T(t)$ der Zeit t , aber im Experiment wird ihr Wert nur zu bestimmten Zeitpunkten t_1, t_2, t_3, \dots gemessen. Praktischerweise nehmen wir an, dass diese Zeitpunkte alle den gleichen Zeitabstand Δt (hier z.B. 1 Minute) haben.

Aus dem Defizit der praktischen Messung, nur Temperaturmesswerte T_i für diese diskrete Zeitpunkte zu liefern, kann man die

Idee beziehen, auch bei der Modellierung diskret zu arbeiten. Die Anfangstemperatur T_0 ist vorgegeben. Wie verändert sich die Temperatur vom i -ten Zeitschritt zum $i+1$ -ten Zeitschritt? Sie wird kleiner, also: $T_{i+1} = T_i - \text{Abkühlung}$.

Die Abkühlung ist umso größer, ...

- je länger die Zeit Δt von t_i bis t_{i+1} ist.
- je größer die Temperaturdifferenz zur Umgebungstemperatur T_U ist.
- je größer die Oberfläche des Körpers und je besser seine Wärmeabstrahlung ist.

Zum letzten Punkt haben wir keine verwertbaren Informationen und versuchen daher einfach, ihn durch eine Konstante c zu berücksichtigen. Weiter nehmen wir mal an, dass die je-desto-Beziehungen aus den obigen drei Punkten alle sogar Proportionalitäten sind. Jetzt müssen die Schüler über folgendes Modellierungsprinzip verfügen,

das im Physikunterricht häufig angewendet wird, in der Mathematik aber eigentlich nicht thematisiert wird: Falls eine Funktion mehrerer Veränderlicher in jeder Variablen eine Proportionalität ist, wenn man die anderen Variablen konstant hält, dann ist die Funktion gleich dem Produkt aller Variablen mit einer Proportionalitätskonstante. Angewendet auf die Abkühlung zwischen den Zeitschritten, bedeutet das, dass die Abkühlung ein Produkt der Zeitdifferenz und der Temperaturdifferenz zur Umgebungstemperatur und der Konstanten sein muss. Damit ergibt sich insgesamt:

$$T_{i+1} = T_i - \Delta t \cdot (T_i - T_U) \cdot c$$

In dieser Gleichung treten zwei Konstanten auf, die man zu einer einzigen zusammenfassen kann ohne das Modell zu beschränken. $C = \Delta t \cdot c$: In dieser Form, also als $T_{i+1} = T_i - (T_i - T_U) \cdot C$ kann das Modell leicht in eine Tabellenkalkulation eingegeben werden:

| | A | B | C | D |
|----|--------|------------|--------------|------|
| 1 | Zeit t | Temperatur | Konstante C= | 0,05 |
| 2 | | 0 | Umgebung= | 20 |
| 3 | | 1 | | |
| 4 | | 2 | | |
| 5 | | 3 | | |
| 6 | | 4 | | |
| 7 | | 5 | | |
| 8 | | 6 | | |
| 9 | | 7 | | |
| 10 | | 8 | | |
| 11 | | 9 | | |
| 12 | | 10 | | |
| 13 | | 11 | | |
| 14 | | 12 | | |
| 15 | | 13 | | |
| 16 | | 14 | | |
| 17 | | 15 | | |
| 18 | | | | |

Die Formeln in den Zellen sind dann:

| | A | B | C | D |
|----|--------|----------------------|--------------|------|
| 1 | Zeit t | Temperatur | Konstante C= | 0,05 |
| 2 | 0 | 80 | Umgebung= | 20 |
| 3 | 1 | =B2-(B2-D\$2)*D\$1 | | |
| 4 | 2 | =B3-(B3-D\$2)*D\$1 | | |
| 5 | 3 | =B4-(B4-D\$2)*D\$1 | | |
| 6 | 4 | =B5-(B5-D\$2)*D\$1 | | |
| 7 | 5 | =B6-(B6-D\$2)*D\$1 | | |
| 8 | 6 | =B7-(B7-D\$2)*D\$1 | | |
| 9 | 7 | =B8-(B8-D\$2)*D\$1 | | |
| 10 | 8 | =B9-(B9-D\$2)*D\$1 | | |
| 11 | 9 | =B10-(B10-D\$2)*D\$1 | | |
| 12 | 10 | =B11-(B11-D\$2)*D\$1 | | |
| 13 | 11 | =B12-(B12-D\$2)*D\$1 | | |
| 14 | 12 | =B13-(B13-D\$2)*D\$1 | | |
| 15 | 13 | =B14-(B14-D\$2)*D\$1 | | |
| 16 | 14 | =B15-(B15-D\$2)*D\$1 | | |
| 17 | 15 | =B16-(B16-D\$2)*D\$1 | | |
| 18 | | | | |

Der Wert von C muss so gewählt werden, dass die berechneten Werte sinnvoll sind (er ist ein Modellparameter). Unabhängig von seinem konkreten Wert führt das Modell aber auf eine exponentielle Temperaturgleichung zur Umgebung. Der Vergleich mit den Messwerten der Kaffeetemperatur aus der obigen Tabelle zeigt eine gute, aber keine perfekte Passung. Hier kann physikalische Modellkritik einsetzen: Bei Flüssigkeiten ist die Abkühlung anfangs stärker, weil zur Wärmeabstrahlung und Wärmeleitung noch die Abkühlung durch Verdunstung hinzu kommt. Dieser Effekt ist so stark, dass die folgende methodische Alternative nur empfohlen werden kann, wenn man tatsächlich die Temperatur von Klötzen aus Metall misst, oder wenn man z.B. durch einen Deckel wie auf den Coffee-to-go-Bechern, die Verdunstung minimiert. In diesem Falle ist die Abkühlung so gut exponentiell, dass man folgendes Vorgehen wählen kann: Man misst (besser in 2 oder 3 Minuten-Abständen) auf $1/10^\circ\text{C}$ genau, damit die Änderung deutlicher ist. In der Tabellenkalkulation werden die Messwerte eingetragen. Inspektion zeigt, dass die Abkühlung, also die Temperaturdifferenz von einem Schritt zum nächsten, kleiner wird. In einer Spalte kann man diese Differenzen von der Tabellenkalkulation ausrechnen

lassen. Die Differenz zur Umgebungstemperatur kann ebenfalls leicht ausgerechnet werden und wenn man die Abkühlung über diese Differenz graphisch aufträgt, ergibt sich näherungsweise eine Ursprungsgerade. Damit ist die oben theoretisch postulierte Proportionalität empirisch abgesichert. Nachteil ist allerdings, dass die notwendige Auftragung für Schüler nicht einleuchtend ist, diese würden eher die Abkühlung über der Zeit auftragen.

3 Ein heißer und ein kalter Klotz

Der Klotz im vorherigen Beispiel hatte eine einheitliche Temperatur. Jetzt nähern wir uns realen Körpern an, die an unterschiedlichen Stellen unterschiedlich heiß sein können – man denke an den Stil einer Pfanne. In ihnen kann die Wärme von einer Stelle zu einer anderen geleitet werden.

Um die Gesetze der Wärmeleitung zu verstehen, nehmen wir erst mal an, wir hätten zwei Eisenklötze. Der eine wird in kochendem Wasser auf 100°C erhitzt, der andere in Eiswasser bis auf 0°C abgekühlt. Danach werden beide auf eine möglichst isolierende Unterlage gelegt und in direkten Kontakt gebracht. Die Temperatur der beiden Klötze (in ihrer Mitte) wird im zeitlichen Verlauf gemessen. Anders als im vor-

herigen Abschnitt soll die Angleichung an die Umgebung zunächst mal vernachlässigt werden. Experimentell bedeutet das, dass man große Klötze nimmt, die sich im Kontakt mit der Luft nicht so schnell abkühlen.

Schüler können aus ihrer Intuition heraus qualitativ skizzieren, wie die Graphen der Temperaturkurven aussehen, sprich, dass sich der eine Körper erwärmt, während der andere sich abkühlt und dass sich etwa 50°C (wenn die Abkühlung an der Luft wirklich vernachlässigt werden kann) einstellen. Diese Annäherung ist anfangs schneller als am Ende. Wir sind also in einer Situation, die der obigen analog ist.

Der Temperatenausgleich geht schneller, es wird also mehr Wärmeenergie übertragen, wenn die Temperaturdifferenz zum Nachbarklotz groß ist. Es bietet sich also eine Formel an der Bauart:

Neue Temperatur = alte Temperatur + Temperaturdifferenz*Konstante

$$\text{Formaler: } T_{\text{neu}} = T_{\text{alt}} + (T_{\text{Nachbar}} - T_{\text{alt}}) \cdot k$$

Die Konstante wählt man nach „Gefühl“. So entsteht das folgende Arbeitsblatt einer Tabellenkalkulation mit $k=1/20$.

| Temperatenausgleich | | | Temperatenausgleich | | |
|---------------------|---------|--------|---------------------|-------------------|-------------------|
| Zeit t | Klotz 1 | Klotz2 | Zeit t | Klotz 1 | Klotz2 |
| 0 | 100,0 | 0,0 | 0 | 100 | 0 |
| 1 | 95,0 | 5,0 | 1 | =B4+(C4-B4)/20 | =C4+(B4-C4)/20 |
| 2 | 90,5 | 9,5 | 2 | =B5+(C5-B5)/20 | =C5+(B5-C5)/20 |
| 3 | 86,5 | 13,6 | 3 | =B6+(C6-B6)/20 | =C6+(B6-C6)/20 |
| 4 | 82,8 | 17,2 | 4 | =B7+(C7-B7)/20 | =C7+(B7-C7)/20 |
| 5 | 79,5 | 20,5 | 5 | =B8+(C8-B8)/20 | =C8+(B8-C8)/20 |
| 6 | 76,6 | 23,4 | 6 | =B9+(C9-B9)/20 | =C9+(B9-C9)/20 |
| 7 | 73,9 | 26,1 | 7 | =B10+(C10-B10)/20 | =C10+(B10-C10)/20 |
| 8 | 71,5 | 28,5 | 8 | =B11+(C11-B11)/20 | =C11+(B11-C11)/20 |
| 9 | 69,4 | 30,6 | 9 | =B12+(C12-B12)/20 | =C12+(B12-C12)/20 |
| 10 | 67,4 | 32,6 | 10 | =B13+(C13-B13)/20 | =C13+(B13-C13)/20 |
| 11 | 65,7 | 34,3 | 11 | =B14+(C14-B14)/20 | =C14+(B14-C14)/20 |
| 12 | 64,1 | 35,9 | 12 | =B15+(C15-B15)/20 | =C15+(B15-C15)/20 |
| 13 | 62,7 | 37,3 | 13 | =B16+(C16-B16)/20 | =C16+(B16-C16)/20 |
| 14 | 61,4 | 38,6 | 14 | =B17+(C17-B17)/20 | =C17+(B17-C17)/20 |
| 15 | 60,3 | 39,7 | 15 | =B18+(C18-B18)/20 | =C18+(B18-C18)/20 |
| 16 | 59,3 | 40,7 | 16 | =B19+(C19-B19)/20 | =C19+(B19-C19)/20 |
| 17 | 58,3 | 41,7 | 17 | =B20+(C20-B20)/20 | =C20+(B20-C20)/20 |
| 18 | 57,5 | 42,5 | 18 | =B21+(C21-B21)/20 | =C21+(B21-C21)/20 |
| 19 | 56,8 | 43,2 | 19 | =B22+(C22-B22)/20 | =C22+(B22-C22)/20 |

4 Physikalische Grundlagen

Ein Körper ist umso wärmer, je mehr Wärmeenergie in ihm steckt. Erwärmen

bedeutet also die Zufuhr von Wärmeenergie, Abkühlen deren Abfuhr.

Um einen großen Körper zu erwärmen, braucht man mehr Energie als bei einem kleinen. Aber auch das Material hat einen

Einfluss: Ein Kilogramm Holz lässt sich mit weniger Energie z.B. um 10°C erwärmen, als ein Kilogramm Wasser. Beides zusammen fasst die Physik in der Größe „Wärmekapazität“ C des Körpers zusammen. Hat der Körper die Temperatur T steckt in ihm die Wärmeenergie $E=C \cdot T$.

Wenn sich zwei Körper berühren, gibt der wärmere Energie an den kühleren ab, und dadurch gleicht sich ihre Temperatur an. Die Geschwindigkeit des Energieübertrags hängt dabei davon ab, wie eng sich die Körper berühren: Er geht schnell, wenn sie fest aneinander gepresst sind, und langsam wenn sie locker nebeneinander stehen. Außerdem wird umso mehr Energie übertragen, je größer die Temperaturdifferenz ist. Die Inhalte an Wärmeenergie der beiden Körper 1 und 2 werden mit E_1 bzw. E_2 bezeichnet, ihre Temperatur mit T_1, T_2 . Diese Größen sind Funktionen der Zeit.

Der Energieübertrag von Körper 1 auf 2 ist dann:

$$\Delta E_{1 \rightarrow 2} = k_{12} \cdot (T_1 - T_2) \cdot \Delta t$$

Darin sind:

$\Delta E_{1 \rightarrow 2}$: Die von 1 auf 2 übertragene Wärmeenergie. Dieser Wert ist positiv, wenn Körper 2 Energie aufnimmt (weil er kühler ist), und negativ, wenn er Energie abgibt.

k_{12} : Wärmeleitkoeffizient, eine Zahl die angibt, wie gut die Wärme zwischen Körper 1 und 2 ausgetauscht werden kann.

Δt : Die Zeitdauer, in der der Wärmeübertrag stattfindet. Diese Zeit muss so klein, dass sich die Temperaturdifferenz $T_1 - T_2$ nur minimal ändert.

Jetzt nehmen wir an, dass beide Körper die gleiche Wärmekapazität C haben. Wenn zu Beginn eines Zeitintervalls Δt die Temperaturen $T_{1,\text{alt}}$ und $T_{2,\text{alt}}$ sind, berechnen sich die neuen Temperaturen folgendermaßen:

$$\begin{aligned} E_{2,\text{neu}} &= E_{2,\text{alt}} + \Delta E_{1 \rightarrow 2} = \\ &= E_{2,\text{alt}} + k_{12} \cdot (T_1 - T_2) \cdot \Delta t \end{aligned}$$

$$\begin{aligned} T_{2,\text{neu}} &= \frac{1}{C} E_{2,\text{neu}} = \\ &= \frac{1}{C} (E_{2,\text{alt}} + k_{12} \cdot (T_1 - T_2) \cdot \Delta t) \\ &= T_{2,\text{alt}} + \frac{k_{12}}{C} \cdot (T_1 - T_2) \cdot \Delta t \end{aligned}$$

Für den anderen Körper gilt analog:

$$T_{1,\text{neu}} = T_{1,\text{alt}} + \frac{k_{12}}{C} \cdot (T_2 - T_1) \cdot \Delta t$$

Damit sind die Berechnungsformeln, die wir in der obigen Excel-Tabelle verwendet haben, theoretisch fundiert. Die Konstanten

waren fest gewählt: $\frac{k_{12}}{C} = \frac{1}{20}, \Delta t = 1$.

5 Wärmeleitung in einem Stab

Mit dem jetzt gewonnen Verständnis der Physik lässt sich auch eine komplexere Fragestellung verstehen: Wie breitet sich die Wärme z.B. im Stil eines Topfes aus? Der Stil soll anfangs kalt sein (Umgebungstemperatur von 20°C), und an einer Seite mit dem heißen Kochgut in Berührung stehen.

Wir zerteilen den Stil dazu in Gedanken in eine bestimmte Anzahl n von Zellen, denen wir eine einheitliche Temperatur unterstellen:

| | | | | |
|-------|-------|-------|-----|-------|
| T_1 | T_2 | T_3 | ... | T_n |
|-------|-------|-------|-----|-------|

In jedem Zeitschritt ändert sich die Temperatur einer Zelle aufgrund der Energieüberträge, die sie vom linken und vom rechten Nachbarn bekommt.

Betrachten wir die Energieänderung des i -ten Segmentes. Diese setzt sich zusammen aus der Energie, die es vom linken Nachbar bekommt und derjenigen vom rechten Nachbar. Die Energiemengen richten sich nach der Temperaturdifferenz und von einer Konstante k , die die Wärmeleitfähigkeit angibt:

$$\Delta E_i = k \cdot (T_{i+1} - T_i) \cdot \Delta t + k \cdot (T_{i-1} - T_i) \cdot \Delta t.$$

Wenn wir jetzt die zeitliche Veränderung betrachten, brauchen wir noch einen zweiten Index j für die Nummer des Zeitschritts.

Die Änderung der Energie wiederum bestimmt, wie im vorigen Abschnitt gezeigt, über die Wärmekapazität eines Segmentes C die Änderung der Temperatur hin zum nächsten Zeitschritt: $T_{i,j+1} = T_{i,j} + \frac{1}{C} \cdot \Delta E_i$.

Wenn man das zusammensetzt und den Index für die Zeit wieder hinschreibt erhält man mit der Abkürzung $c = \frac{k}{C} \cdot \Delta t$ die Gleichung:

$$T_{i,j+1} - T_{i,j} = c \cdot (T_{i+1,j} - 2T_{i,j} + T_{i-1,j})$$

Dies ist eine einfache kausale Formel, denn sie bestimmte den Wert im nächsten Zeitschritt $j+1$ allein aus Größen des Zeitschritts j .

Im Tabellenkalkulationsblatt wird die Zeit nach unten zunehmen, und die Rechtsdimension ist die Ausdehnung des Stabes. Mit dieser Wahl überlegen wir uns jetzt den Aufbau der Zelle eines Blattes:

| Position | $i=1$ | $i=2$ | $i=3$ | $i=4$ | $i=5$ |
|------------|-----------|-----------|-----------|-----------|-----------|
| Zeit $t=0$ | $T_{1,0}$ | $T_{2,0}$ | $T_{3,0}$ | $T_{4,0}$ | $T_{5,0}$ |
| Zeit $t=1$ | $T_{1,1}$ | $T_{2,1}$ | $T_{3,1}$ | $T_{4,1}$ | $T_{5,1}$ |
| Zeit $t=2$ | | | | | |

Angenommen, es sind schon alle Zellen berechnet, in denen $T_{i,j}$ steht. Welche Formel muss in die graue Zelle? Aus $T_{i,j+1} - T_{i,j} = c \cdot (T_{i+1,j} - 2T_{i,j} + T_{i-1,j})$ ergibt sich in diesem Spezialfall: $T_{3,2} = T_{3,1} + c \cdot (T_{4,1} - 2T_{3,1} + T_{2,1})$.

Mit diesen Gleichungen können wir also aus den Temperaturen einer Zelle und ihrer Nachbarzellen in einem Zeitschritt die Temperatur im nächsten Zeitschritt ausräumen. Was aber ist mit den Zellen am Rand des Stabes, die haben ja keine linke, bzw keine rechte Nachbarzelle? Ihre Temperatur wird fest gesetzt: Der linke Rand soll z.B. immer Umgebungstemperatur haben, der rechte Rand wird z.B. auf 100°C festgesetzt. Diese willkürlich gesetzten Zahlen sind im folgenden Tabellenblatt grau eingefärbt. Das gesamte Tabellenblatt sieht dann so aus:

| | A | B | C | D | E | F | G | H | I | J | K |
|----|---|------------|------|------|------|------|------|------|------|------|-------|
| 1 | Wärmeleitung | | | | | | | | | | |
| 2 | | | | | c= | 0,5 | | | | | |
| 3 | | | | | | | | | | | |
| 4 | | | | | | | | | | | |
| 5 | Graue Zwillkürliche gewähle Bedingungen | | | | | | | | | | |
| 6 | Zeit t | Positionen | | | | | | | | | |
| 7 | 0 | 1 | 2 | 3 | 4 | 5 | 6 | 7 | 8 | 9 | 10 |
| 8 | 1 | 20,0 | 20,0 | 20,0 | 20,0 | 20,0 | 20,0 | 20,0 | 20,0 | 20,0 | 100,0 |
| 9 | 2 | 20,0 | 20,0 | 20,0 | 20,0 | 20,0 | 20,0 | 20,0 | 20,0 | 60,0 | 100,0 |
| 10 | 3 | 20,0 | 20,0 | 20,0 | 20,0 | 20,0 | 20,0 | 20,0 | 40,0 | 60,0 | 100,0 |
| 11 | 4 | 20,0 | 20,0 | 20,0 | 20,0 | 20,0 | 20,0 | 30,0 | 40,0 | 70,0 | 100,0 |
| 12 | 5 | 20,0 | 20,0 | 20,0 | 20,0 | 20,0 | 25,0 | 30,0 | 50,0 | 70,0 | 100,0 |
| 13 | 6 | 20,0 | 20,0 | 20,0 | 20,0 | 22,5 | 25,0 | 37,5 | 50,0 | 75,0 | 100,0 |
| 14 | 7 | 20,0 | 20,0 | 20,0 | 21,3 | 22,5 | 30,0 | 37,5 | 56,3 | 75,0 | 100,0 |
| 15 | 8 | 20,0 | 20,0 | 20,6 | 21,3 | 25,6 | 30,0 | 43,1 | 56,3 | 78,1 | 100,0 |
| 16 | 9 | 20,0 | 20,3 | 20,6 | 23,1 | 25,6 | 34,4 | 43,1 | 60,6 | 78,1 | 100,0 |
| 17 | 10 | 20,0 | 20,3 | 21,7 | 23,1 | 28,8 | 34,4 | 47,5 | 60,6 | 80,3 | 100,0 |
| 18 | 11 | 20,0 | 20,9 | 21,7 | 25,2 | 28,8 | 38,1 | 47,5 | 63,9 | 80,3 | 100,0 |
| 19 | 12 | 20,0 | 20,9 | 23,0 | 25,2 | 31,7 | 38,1 | 51,0 | 63,9 | 82,0 | 100,0 |
| 20 | 13 | 20,0 | 21,5 | 23,0 | 27,4 | 31,7 | 41,3 | 51,0 | 66,5 | 82,0 | 100,0 |
| 21 | 14 | 20,0 | 21,5 | 24,4 | 27,4 | 34,4 | 41,3 | 53,9 | 66,5 | 83,2 | 100,0 |
| 22 | 15 | 20,0 | 22,2 | 24,4 | 29,4 | 34,4 | 44,1 | 53,9 | 68,6 | 83,2 | 100,0 |
| 23 | 16 | 20,0 | 22,2 | 25,8 | 29,4 | 36,8 | 44,1 | 56,4 | 68,6 | 84,3 | 100,0 |
| 24 | 17 | 20,0 | 22,9 | 25,8 | 31,3 | 36,8 | 46,6 | 56,4 | 70,3 | 84,3 | 100,0 |
| 25 | 18 | 20,0 | 22,9 | 27,1 | 31,3 | 38,9 | 46,6 | 58,4 | 70,3 | 85,2 | 100,0 |

Man sieht an diesen Werten schön, wie sich die Wärme vom rechten Rand aus in den Stab hinein ausbreitet. Die Formeln schauen wir uns nur im Bereich von A7 bis

D10 an, denn sie sind – wie üblich – mit der Ausfüllfunktion der Tabellenkalkulation leicht erzeugbar:

| | | | |
|---|----|-------------------------|-------------------------|
| 0 | 1 | 2 | 3 |
| 1 | 20 | 20 | 20 |
| 2 | 20 | =C8+\$F\$2*(B8+D8-2*C8) | =D8+\$F\$2*(C8+E8-2*D8) |
| 3 | 20 | =C9+\$F\$2*(B9+D9-2*C9) | =D9+\$F\$2*(C9+E9-2*D9) |

6 Abschließende Bemerkungen

Die Tabellenkalkulation erlaubt, ein nicht-triviales Problem rechnerisch approximativ zu lösen. Schüler können dabei erkennen, dass die Natur dank ihrer mathematischen Modellierung berechenbar ist.

Obwohl sich die Komplexität der Tabelle in Grenzen hält, leistet sie doch viel. Von einem mathematischen Standpunkt aus wurde hier die partielle Differentialgleichung

$\frac{\partial T}{\partial t} = c \cdot \frac{\partial^2 T}{\partial x^2}$, die so genannte Wärmeleitungsgleichung, näherungsweise gelöst.

